VHDL testbänk

Mall-programmets funktion



Låset öppnas när tangenten ”1” trycks ned och sedan släpps.



William Sandqvist william@kth.se

Keypad och Statecounter

Bra val av datatyper gör koden själv-förklarande!






```
K: in  std_logic_vector(1 to 3);
R: in  std_logic_vector(1 to 4);
```

1 2 3

K ="001" bitvector

K(3)='1' bit

1 2 3 4

R ="0001" bitvector

R(4)='1' bit

4 3 2 1 0

Statecounter: **q ="00001"** bitvektor

q(0)='1' bit

William Sandqvist william@kth.se

lockmall.vhd

This code is given


```
library IEEE;
use IEEE.std_logic_1164.all;
use IEEE.std_logic_arith.all;

entity codelock is
   port( clk:      in  std_logic;
         K:      in  std_logic_vector(1 to 3);
         R:      in  std_logic_vector(1 to 4);
         q:      out std_logic_vector(4 downto 0);
         UNLOCK: out std_logic );
end codelock;

architecture behavior of codelock is
subtype state_type is integer range 0 to 31;
signal state, nextstate: state_type;

begin
nextstate_decoder: -- next state decoding part
process(state, K, R)
 begin
   case state is
      when 0 => if (K = "100" and R ="0001")     then nextstate <= 1;
                else nextstate <= 0;
                end if;
      when 1 => if (K = "100" and R ="0001")     then nextstate <= 1;
                elsif (K = "000" and R = "0000") then nextstate <= 2;
                else nextstate <= 0;
                end if;
      when 2 to 30 => nextstate <= state + 1;
      when 31      => nextstate <= 0;
   end case;
end process;

debug_output:  -- display the state
q <= conv_std_logic_vector(state,5);
```

```
output_decoder: -- output decoder part
process(state)
begin
  case state is
     when 0 to 1  => UNLOCK <= '0';
     when 2 to 31 => UNLOCK <= '1';
  end case;
end process;

state_register: -- the state register part (the flipflops)
process(clk)
begin
  if rising_edge(clk) then
     state <= nextstate;
  end if;
end process;
end behavior;
```

It's easy to see that this is correct!



William Sandqvist william@kth.se

lockmall_with_error.vhd

```
library IEEE;
use IEEE.std_logic_1164.all;
use IEEE.std_logic_arith.all;

entity codelock is
   port( clk:      in  std_logic;
         K:      in  std_logic_vector(1 to 3);
         R:      in  std_logic_vector(1 to 4);
         q:      out std_logic_vector(4 downto 0);
         UNLOCK: out std_logic );
end codelock;

architecture behavior of codelock is
subtype state_type is integer range 0 to 31;
signal state, nextstate: state_type;

begin
nextstate_decoder: -- next state decoding part
begin
nextstate_decoder: -- next state decoding part
process(state, K, R)
 begin
   case state is
      when 0 => if(((R(2)='0') and (R(3)='0') and (K(2)='0') and (K(3)='1')) and
                   ( not (( not ((K(1)='0') and (R(1)='0') and (R(4)='1'))) and
                   ( not ((K(1)='1') and (R(1)='1') and (R(4)='0'))))))
                then nextstate <= 1;
                else nextstate <= 0;
                end if;
      when 1 => if(((R(2)='0') and (R(3)='0') and (K(2)='0') and (K(3)='1')) and
                   ( not (( not ((K(1)='0') and (R(1)='0') and (R(4)='1'))) and
                   ( not ((K(1)='1') and (R(1)='1') and (R(4)='0'))))))
                            then nextstate <= 1;
                elsif (K = "000" and R = "0000") then nextstate <= 2;
                else nextstate <= 0;
                end if;
      when 2 to 30 => nextstate <= state + 1;
      when 31      => nextstate <= 0;
   end case;
end process;
```

```
debug_output:  -- display the state
q <= conv_std_logic_vector(state,5);

output_decoder: -- output decoder part
process(state)
begin
  case state is
     when 0 to 1  => UNLOCK <= '0';
     when 2 to 31 => UNLOCK <= '1';
  end case;
end process;

state_register: -- the state register part (the flipflops)
process(clk)
begin
  if rising_edge(clk) then
     state <= nextstate;
  end if;
end process;
end behavior;
```

Now it's hard to see if this is correct or not?

William Sandqvist william@kth.se

lockmall_with_error.vhd

Betyder båda uttrycken samma sak?

```
( K = "100" and R ="0001" )
```

Är verkligen detta samma sak?

```
(((R(2)='0') and (R(3)='0') and (K(2)='0') and (K(3)='1')) and
( not (( not ((K(1)='0') and (R(1)='0') and (R(4)='1'))) and
( not ((K(1)='1') and (R(1)='1') and (R(4)='0'))))))
```

Någon ”lovar” att koden är korrekt – men hur kan man veta att detta är absolut sant?

William Sandqvist william@kth.se

Testbench

thank’s to: *Francesco Robino*

`tb_lockmall.vhd`

William Sandqvist william@kth.se

tb_lockmall.vhd

Vi behöver skriva en VHDL-testbench

Ett testbänksprogram kan testa alla möjliga tangentkombinationer och rapportera om det uppstår något problem …

Det kan automatiskt loopa igenom all möjliga tangenttryckningar och rapportera om om låset försöker att öppna.

Det finns $2^7 = 128$ möjliga tangentkombinationer och vi skulle bli helt uttröttade om vi försökte att prova dem alla för hand.

`entity` – en testbänk har inga portar

```
entity tb_codelock is
 -- entity tb_codelock has no ports
 -- because it's for simulation only
end tb_codelock;
```

Några interna signaler behövs

```
signal         clk : std_logic := '0';
signal      K_test : std_logic_vector(1 to 3);
signal      R_test : std_logic_vector(1 to 4);
signal prev_K_test : std_logic_vector(1 to 3);
signal prev_R_test : std_logic_vector(1 to 4);
signal           q : std_logic_vector(4 downto 0);
signal      unlock : std_logic;
```

Vårt `codelock` används som component

```
-- we use our codelock as a component
component codelock
  port( clk : in std_logic;
          K : in std_logic_vector(1 to 3);
          R : in std_logic_vector(1 to 4);
          q : out std_logic_vector(4 downto 0);
     UNLOCK : out std_logic );
end component;
```

Genera en simuleringsklocka




```
-- generate a simulation clock
clk <= not clk after 10 ns;
```

Instantiatiering och signal mapping

```
-- instantiation of the device under test,
-- mapping of signals
inst_codelock:
  codelock
  port map (
              clk => clk,
                K => K_test,
                R => R_test,
                q => q,
           UNLOCK => unlock );
```

En nästlad slinga skapar tangenttryckningarna




```
process
begin
 for k in 0 to 7 loop
   K_test <= conv_std_logic_vector(k,3);
   for r in 0 to 15 loop
     prev_K_test <= K_test;
     prev_R_test <= R_test;
     R_test <= conv_std_logic_vector(r,4);
     wait until CLK='1';
   end loop;
 end loop;
end process;
```

8·16=128 turns

William Sandqvist william@kth.se

report, severity note, severity error

Tests if state q = ”00001” will be reached by any combination.




```
check:
process(q)
begin if ((q = "00001") and
          (prev_K_test = conv_std_logic_vector(1,3)) and
          (prev_R_test = conv_std_logic_vector(1,4)))
       then assert false report
         "Lock tries to open for the right sequence!”
         severity note;
       else if ((q = "00001"))
       then
        assert false report
         "Lock tries to open with the wrong sequence!”
         severity error;
       else report "Lock closed!" severity note;
            end if;
     end if;
 end process check;
```

William Sandqvist william@kth.se

Simulera och hitta felet!

Vad annat än att trycka på ”1” tangenten skulle kunna öppna låset?

?

William Sandqvist william@kth.se